

Limits at Infinity Plus More Advanced Methods for Calculus and Graphics

Introduction and Goals

This lab will get into some more advanced techniques, as well as revisit some old ones, for graphing and analyzing functions. Its main purpose is to show you a little more of the power of Maple and how to make certain manipulations easier.

Before You Start

Before you start you may want to review the Maple commands in the previous labs.

Textbook Correspondence

Stewart 5th edition: Many different sections.

Maple Commands and Packages Used

Packages: plots

Commands: solve, subs, display, plot, seq, diff, limit, quo, piecewise.

Maple Commands

The subs Command:

Usually when we are working with Maple we will define a function like

```
> f := x -> x^2;
```

$$f := x \rightarrow x^2$$

and then if we wish to substitute a value or expression in for x we simply use function notation, as in the examples below.

```
> f(2);
```

4

```
> f(a+h);
```

$$(a+h)^2$$

In previous labs we noted that we could not substitute values in for x if we had an expression and not a function. Maple does have a way to substitute values or expressions into another expression, the subs command. For example, the command

> **subs (x=2, x^2+3*x-1) ;**

9

substitutes 2 in for x in the given expression. The command

> **subs (x=a+h, x^2+3*x-1) ;**

$(a+h)^2 + 3a + 3h - 1$

substitutes $a+h$ in for x in the given expression and the command

> **subs (x=sin(x), x^2+3*x-1) ;**

$\sin(x)^2 + 3\sin(x) - 1$

substitutes $\sin(x)$ in for x in the given expression.

The subs command does have its limitations, for example,

> **subs (x-1=sin(x), x^2+3*x-1) ;**

$x^2 + 3x - 1$

Note that it made no substitutions. On the other hand had we used a more powerful substitution command, `algsubs`, we would have received,

> **algsubs (x-1=sin(x), x^2+3*x-1) ;**

$3 + 5\sin(x) + \sin(x)^2$

In the above command Maple rewrote

$x^2 + 3x - 1$

as

$(x-1)^2 + 5(x-1) + 3$

and then did the substitution. To see the equivalence try the commands.

> **(x-1)^2+5*(x-1)+3;**

$(x-1)^2 + 5x - 2$

> **expand(%);**

$x^2 + 3x - 1$

In general, it is easier to use function instead of the subs command but there are times when the subs command comes in handy. For example, in creating a general tangent line function.

A General Tangent Line Function:

In this section we will see how we can use Maple to create a general tangent line function. This function will create a tangent line to any curve at any point on the curve and furthermore do it as a function. Consider the command,

```
> TL := (a, x) -> subs (t=a, diff (f (t) , t) ) * (x-a) + f (a) ;
```

$$TL := (a, x) \rightarrow \text{subs} \left(t = a, \frac{d}{dt} f(t) \right) (x - a) + f(a)$$

What it is doing is substituting the value of a into the derivative of $f(x)$ and placing that slope into the point slope form of the equation of a line. Thus creating the tangent line to the curve at a . Two things to note about this command are first, it creates a tangent line to $f(x)$, so f must be defined as a function before using the command. Second, the function only needs to be defined once. When you change the f function the tangent line equation automatically changes.

For example, assuming that the TL function command has been input then

```
> f := x -> x^2;
```

$$f := x \rightarrow x^2$$

```
> TL (2, x) ;
```

$$4x - 4$$


creates the tangent line to $f(x) = x^2$ at $x = 2$. Furthermore, if we wanted to know the y value of this tangent line when $x = 5$ we would use the command.

```
> TL (2, 5) ;
```

$$16$$

Finally, to plot the function and its tangent line we would write.

```
> plot ([f (x) , TL (2, x) ] , x=-5..5, y=-10..10) ;
```




As was pointed out above, simply by changing the f function the TL function automatically changes. For example,

```
> f:=x->sin(x);
```

$f := \sin$


```
> plot([f(x), TL(2, x)], x=-5..5, y=-3..3);
```



```
> f:=x->log[3](x);
```

$f := \log_3$

```
> plot([f(x), TL(2, x)], x=-5..5, y=-3..3);
```



Animation:

Another more advanced feature of Maple is its ability to produce animations. To do animations and some other more advanced graphics we need to load the plots package into our session. Maple has many built-in functions that load when you first start the program, but it does not load all of its functionality. If it did, many computers would run out of memory. So instead, the more advanced features of Maple are stored in packages that are loaded when you need them. The one we will need in this lab is the plots package. All packages are loaded using the with command.

```
> with(plots) :
```

```
Warning, the name changecoords has been redefined
```

Note that we used a colon instead of a semicolon. To produce an animation we construct a sequence of plots and then paste them together with the display command. For example, redefine the f function as


```
> f:=x->x^2;
```

$$f := x \rightarrow x^2$$

Now to animate the tangent line moving down the function we use the command,

```
> display(seq(plot([f(x), TL(a, x)], x=-5..5, y=-10..25), a=-5..5), insequence=true);
```

Here we are creating a sequence of 11 plots ($a = -5, -4, -3, -2, -1, 0, 1, 2, 3, 4, 5$) and then pasting them together using the display command with the insequence=true option. On the screen you will see the following image.



Click on the image and look in the toolbar. Instead of the normal icons you will see icons that look more like CD player controls. Press the Play key to start the animation. You will note that this animation is a bit quick. To slow it down we can simply add more frames by increasing the range on a and then dividing a by 10 in the TL function, as is done below.

```
> display(seq(plot([f(x), TL(a/10, x)], x=-5..5, y=-10..25), a=-50..50), insequence=true);
```

Try this command in Maple and run the animation. This animation uses 101 frames.

Exploring Asymptotes A Bit Further:

In class we have looked at vertical and horizontal asymptotes. Recall that vertical asymptotes occurred when a limit was infinite and horizontal asymptotes were when limits at infinity existed. In Maple, a limit at infinity is done like any other limit except that instead of placing $x = 2$ in the limit we place $x = \text{infinity}$ or $x = -\text{infinity}$ in the limit. For example,

```
> limit(arctan(x), x=infinity);
```

$$\frac{\pi}{2}$$

```
> limit(arctan(x), x=-infinity);
```

$$-\frac{\pi}{2}$$

```
> limit((2*x^2-3*x+4)/(x^2-5*x+9), x=infinity);
```

$$2$$

```
> limit(x^2, x=infinity);
```

$$\infty$$

In general, an asymptote is a line or curve that a function approaches. For example, the function


$$\frac{4x^3 - 3x^2 + 17x - 5}{x^2 + x + 1}$$

gets close to

$$4x - 7$$

as the value of x increases or decreases. This can easily be seen in the graph,

`> plot ([f(x), 4*x-7], x=-10..10);`



We will only consider asymptotes to rational functions. In our above example note that

$$\frac{4x^3 - 3x^2 + 17x - 5}{x^2 + x + 1} = 4x - 7 + \frac{2(1 + 10x)}{x^2 + x + 1}$$

The rational function

$$\frac{1 + 10x}{x^2 + x + 1}$$

goes to 0 as the value of x increases or decreases. Hence as the value of x increases or decreases

$$\frac{4x^3 - 3x^2 + 17x - 5}{x^2 + x + 1}$$

gets close to

$$4x - 7$$

For rational functions finding the asymptotic function is easy, note that

$$4x - 7$$

is simply the quotient when you do the long division of

$$\frac{4x^3 - 3x^2 + 17x - 5}{x^2 + x + 1}$$

In Maple the quotient is found using the quo command. For example,

```
> quo (4*x^3-3*x^2+17*x-5, x^2+x+1, x);  
4x - 7
```

As another example,


```
> f:=x->(x^4-10*x^3+3*x^2-7*x+9)/(2*x^2+9*x-5);
```

$$f := x \rightarrow \frac{x^4 - 10x^3 + 3x^2 - 7x + 9}{2x^2 + 9x - 5}$$

```
> quo (x^4-10*x^3+3*x^2-7*x+9, 2*x^2+9*x-5, x);
```

$$\frac{1}{2}x^2 - \frac{29}{4}x + \frac{283}{8}$$

```
> plot ([f(x), 1/2*x^2-29/4*x+283/8], x=-20..30, y=-400..400, discontin=true);
```



Notice the use of the option `discontin=true`. Recall that this command removes the graphing of vertical asymptotes if it can.

Piecewise Defined Functions:

Maple also has the ability to construct piecewise defined functions. Although we do not tend to look at piecewise defined functions very much in Calculus they play an important role in other areas of mathematics. To create a piecewise defined function we simply use the `piecewise` command. In the `piecewise` command you put in a condition followed by the function to use for that condition. The last entry is a default function that is used when none of the conditions are met. For example,

```
> piecewise( x<0, 1, x<1, x, x<2, 3-x, sin(x)+2);
```


$$\begin{cases} 1 & x < 0 \\ x & x < 1 \\ 3-x & x < 2 \\ \sin(x)+2 & \text{otherwise} \end{cases}$$

We can also define a function as a piecewise function. The only difference is that the output format is not as nice.

```
> f:=x->piecewise( x<0, 1, x<1, x, x<2, 3-x, sin(x)+2);  
f := x → piecewise(x < 0, 1, x < 1, x, x < 2, 3 - x, sin(x) + 2)
```

Maple can plot piecewise functions simply with the `plot` command. You should always use the `discont=true` option when plotting piecewise functions.

```
> plot( f(x), x=-2..5, y=0..3, discont=true);
```



One thing to note about the conditions in a piecewise defined function is that if you want to input a range you need to use “and”. Maple will not understand $0 < x < 2$.


```
> piecewise( x<0, 1, x>=0 and x<2, -x+1, x>=2 and x<4, -1,  
x>=4 and x<6, x-5, 1);
```

$$\begin{cases}
 1 & x < 0 \\
 1-x & -x \leq 0 \text{ and } x < 2 \\
 -1 & 2-x \leq 0 \text{ and } x < 4 \\
 x-5 & 4-x \leq 0 \text{ and } x < 6 \\
 1 & \text{otherwise}
 \end{cases}$$

```
> f:=x->piecewise( x<0, 1, x>=0 and x<2, -x+1, x>=2 and x<4,
-1, x>=4 and x<6, x-5, 1);
```

```
f:=x -> piecewise(x < 0, 1, 0 ≤ x and x < 2, 1 - x, 2 ≤ x and x < 4, -1, 4 ≤ x and x < 6,
x - 5, 1)
```

```
> plot (f(x) , x=-1..10, discontinuity=true);
```




The above example is similar to a homework problem. We were asked to find and graph the derivative of this function. Again, with Maple, this is easy.

```
> diff (f(x) , x);
```

$$\begin{cases}
 0 & x < 0 \\
 \text{undefined} & x = 0 \\
 -1 & x < 2 \\
 \text{undefined} & x = 2 \\
 0 & x < 4 \\
 \text{undefined} & x = 4 \\
 1 & x < 6 \\
 \text{undefined} & x = 6 \\
 0 & 6 < x
 \end{cases}$$

```
> plot (diff (f(x) , x) , x=-1..10, discontinuity=true, thickness=3);
```



We increased the thickness of the line so that the axes would not obscure the graph.

Exercises:

- 1) Consider the following function, Find the tangent line to the curve

$$x^4 - 3x^3 - 7x^2 - 2$$

- Find the tangent line to the curve at $x = -2, 0, 2$ and 4 .
 - Graph the function on $[-5, 5] \times [-100, 10]$ along with each of these tangent lines.
 - Create an animation of the tangent line moving along the curve from $a = -2$ to $a = 4$ using 61 frames. Make sure that include the command you used to create the animation.
- 2) For each of the following functions find all horizontal asymptotes or asymptotic functions. Graph the function along with its asymptotes, use ranges for x and y that display the fact that the function is approaching the asymptote.

$$\frac{4x^3 + 8x^2 - 4x + 26}{2x^2 + x - 2}$$

$$\frac{8x^2 - 4x + 26}{2x^2 + x - 2}$$

$$\frac{-4x + 26}{2x^2 + x - 2}$$

- 3) Use Maple to define the function

$$f(x) = \begin{cases} x^2 - \sin(x) & \text{if } x < 0 \\ x^3 - x^2 + x & \text{if } 0 \leq x < 4 \\ \cos(x) - e^{-x} + \log(2x) & \text{if } 4 \leq x \end{cases}$$

- Graph the function.
- Where is the function discontinuous? Verify the continuity or discontinuity of the function at each piece break point.

- c) Find the derivative of the function.
 d) Graph the function along with its derivative on the same plot.
 4) This exercise is simply an analysis of the function,

$$\frac{x^{\left(\frac{13229}{5000}\right)}}{x^{(\sqrt{7})} - x + 2}$$

- a) Define this function as $f(x)$.
 b) Graph this function on the intervals $[0,100]$, $[0,1000]$, $[0,10000]$ and $[0,\infty)$. Does the function have a horizontal asymptote? If so, where? What does this say about

$$\lim_{x \rightarrow \infty} \frac{x^{\left(\frac{13229}{5000}\right)}}{x^{(\sqrt{7})} - x + 2}$$

- c) Use numeric techniques, with very large values of x , to determine

$$\lim_{x \rightarrow \infty} \frac{x^{\left(\frac{13229}{5000}\right)}}{x^{(\sqrt{7})} - x + 2}$$

- d) Use Maples limit command to find

$$\lim_{x \rightarrow \infty} \frac{x^{\left(\frac{13229}{5000}\right)}}{x^{(\sqrt{7})} - x + 2}$$

- e) Use the techniques discussed in class to determine

$$\lim_{x \rightarrow \infty} \frac{x^{\left(\frac{13229}{5000}\right)}}{x^{(\sqrt{7})} - x + 2}$$

- f) Take all of the above information and discuss your findings. I am expecting several paragraphs of thoughtful comments and observations.